

[7.18] Simplest (?) application launcher

Here's the problem: as I use my 92+ more and more, I write and download more programs. Considering that I keep each application in its own folder, and the 8-character name limit prevents really memorable names, I can't remember where a program was, or what it was called. The problem is solved with an application launcher, which shows intelligible descriptions of the programs, then executes the one I choose.

This tip shows two application launchers. I wrote the first one, which sets the application folder. Daniel Lloyd wrote the second one, which is smaller. Daniel's method does not set the application's folder, which doesn't matter if the application doesn't need it, or if the application sets the folder itself.

The first application launcher

The code below shows a simple application launcher.

```
apps()  
Prgm  
local k,appdesc,appfold  
setfold(main)  
mat▶list(submat(appsdef,1,1,rowdim(appsdef),1)→appdesc  
popup appdesc,k  
appsdef[k,2]→appfold  
setfold(#appfold)  
expr(appfold&"\"&main\appsdef[k,3])  
setfold(main)  
EndPrgm
```

apps() displays a pop-up box from which I select the program by its description. *apps()* then sets the current folder to the application folder, executes the program, then sets the current folder to *main* when the program is done.

The application information is stored as strings in a 3-column matrix named *appsdef*. Use the matrix editor to create and edit this matrix. The three columns are:

column 1: Application description
column 2: Application folder
column 3: Application program name

A typical *appsdef* matrix might look like this:

c1	c2	c3
"Voltage divider solver"	"voltdiv"	"vdiv()"
"Cubic spline"	"cubspline"	"spline()"
"RTD resistance"	"rtdeqs"	"RTD385()"

For example, the program *vdiv()* is located in folder *voltdiv*. The description that will be shown in the pop-up box is "Voltage divider solver".

The second application launcher

Daniel's application launcher looks like this:

```
kbdprgm1()  
Prgm  
setFold(main)  
Local a  
PopUp mat>list(pmat[1]),a  
expr(pmat[2,a])  
EndPrgm
```

Since the program is named *kbdprgm1*, it can be launched from any folder by pressing [DIAMOND] [1]. You could also rename my *apps()* program to *kbdprgm1*, with the same effect.

In this program, the application information is saved in a 2-row matrix called *pmat*. One column is used for each application. For the example applications above, *pmat* looks like this:

c1	c2	c3
"Voltage divider solver"	"Cubic Spline"	"RTD resistance"
"voltdiv\vdiv()"	"cubspline\spline()"	"rtdeqs\RTD385()"

The first row shows the text that will appear in the pop-up menu. The second row specifies the folder and the application name, including the parentheses.

Daniel's launcher requires a simpler matrix to save the application information. Also, since he saves the pop-up menu items as the *row* of the matrix, instead of the column, he can simply use

```
PopUp mat>list(pmat[1]),a
```

to create the pop-up menu, since *pmat[1]* accesses the entire first row of the matrix. Daniel's launcher is also smaller because he doesn't set the folder to *main* after the application executes. This isn't really necessary, but you can add it, if you want it.

(Credit for second method to Daniel Lloyd)